E-Field and Electric Potential for a thick shell of charge with a $\rho[r] = \rho_o$ Exp[$-\alpha$ r]

A thick, non-conducting spherical shell of inner radius *a* and outer radius *b* has a volume charge density

ρ[r] = ρ_o Exp[$-\alpha$ r] for a ≤ r ≤ b; assume ρ_o and α are positive.
ρ[r] = 0 elsewhere (for r < a and r > b).
Find $\vec{E}$ and V at all r.



(a)Just for reference, make a plot of ρ[r] (we need to assume some values for the constants; I'll use a = 1; b = 2; α = 0.5; ρo = 1)

```
ClearAll["`*"]
a = 1; b = 2; α = 0.5; ρo = 1;
rho[r_] = If[r < a, 0, If[a < r < b, ρo Exp[ - α r], If[r > b, 0]]]
Plot[rho[r], {r, 0, 6}, AxesLabel → {"r", "ρ[r]"}]
```

```
If[r < 1, 0, If[a < r < b, ρo Exp[-α r], If[r > b, 0]]]
```



We want to Find $\vec{E}$. Let's not beat around the bushes - we will use Gauss's Law to find $\vec{E}$.

(b) But first, in the text cell below, show that the vector $\vec{E}$ is radial in direction *and* that the magnitude of $\vec{\boldsymbol{E}}$ depends only on the distance r from the origin O, where r = $|\vec{r}|$, the magnitude of $\vec{r}$. Remember r is a distance, therefore a scalar. This allows us to write: $\vec{\boldsymbol{E}}[\,\vec{\boldsymbol{r}}\,] = E[\,r\,]\,\hat{\boldsymbol{r}}$. (Note carefully where the symbols have vector vs. scalar status.) This greatly simplifies applying Gauss's Law to find $\vec{\boldsymbol{E}}$.

this is a text cell – type in your response You might want to include a figure.

Normally we simply state that ρ is spherically symmetric, so that $\vec{E}$ is radial and the magnitude of $\vec{E}$, Emag, depends only on the distance from the origin O, i.e., on the scalar distance r. But how do we actually know this? (Hang on to to your hat!)

First, verify the radial direction of $\vec{E}$. **Although not stated, the above figure is a planar SLICE through the spherical charge distribution, passing though the origin. With some added and subtracted features, this is the same slice:**



We know that if we take a small (differential) volume element, say dτ1' within the shell of charge at the position $\overrightarrow{r1'}$ [These ' indicate coordinates associated with source points - where there is some charge], the volume element will contain a differential charge dq1 = ρ dτ1'. This charge produces a differential $\vec{E}$ field, $d\overrightarrow{E1}$, at the field point P (at position $\vec{r}$, in this case outside of the charged shell), where $d\overrightarrow{E1} = k\,\frac{dq1}{\text{R1}^2}\,\hat{\text{R}}1$, where $\overrightarrow{\text{R1}}$ is the displacement vector between dq1 and P, i.e., $\overrightarrow{\text{R1}} = \vec{r} - \overrightarrow{r1'}$ and $\hat{\text{R}}1$ is the unit vector in the direction of $\overrightarrow{\text{R1}}$.

For a radially symmetric ρ, we can find a unique volume element dτ2' with the same differential

volume and value of ρ, as shown in the diagram. [Since the source vectors $\overrightarrow{r1'}$ and $\overrightarrow{r2'}$ have the same lengt, ρ(r1') = ρ(r2').] Therefore dq1 = dq2. Forming $d\overrightarrow{E2} = k\,\frac{dq1}{R2^2}\,\hat{R}2$ and noting from the symmetry we realize that R1 = R2 (the distances), we conclude that $d\overrightarrow{E1}$ and $d\overrightarrow{E2}$ have equal z components and equal BUT OPPOSITE components perpendicular to z. Therefore their sum, $d\vec{E}$ = $d\overrightarrow{E1} + d\overrightarrow{E2}$ has only a z component. As drawn, the z axis is in the $\hat{r}$ direction, therefore is normal to the spherical surface drawn through P (i.e., with center O and radius r).

All the points on the left side of the 3D charge distribution can be paired with corresponding points on the right side of the distribution, so that when the total $\vec{E}$ is found (by integration), the resulting $\vec{E}$ on the surface of a sphere of radius r will be normal to the spherical surface.

IF you are as tired as I am, you will nod enthusiastically at the following: This process can be applied to P located anywhere (0 ≤ r ≤ ∞, OR saying the same thing: for r < a; a ≤ r ≤ b; and r > b).

What about the magnitude of $\vec{E}$ at the point P being a function of r the scalar r — the length of $\vec{r}$) only?

In spherical coordinates, the variables upon which $|\vec{E}|$ could depend are {r, θ, ϕ}. If you pick a particular point P and then rotate the charge distribution to ANY other θ and ϕ, the symmetry says the distribution of charge is not changed. Thus, $|\vec{E}|$ cannot depend on θ or ϕ. Obviously the r dependence remains. (Think r = 0 vs r ≫ b in our problem.) So the magnitude of $\vec{E}$ depends on r only. (This means that $|\vec{E}|$ is constant on any spherical surface centered at O.

For obvious reasons, ONCE we go through this lengthy exercise (or similar arguments for other symmetries) we generally and happily say "because of symmetry, . . . " and quickly move on.

(c) State why we make these arguments.

```
this is a text cell – type in your response
```

In order to efficiently and correctly apply Gauss's Law. On spherical Gaussian surfaces centered at the origin O, the integrals that arise using Gauss's Law become greatly simplified. In fact, the integrals are usually impossible without these simplifications.

(d) Find $\vec{E}[\vec{r}]$ for r < a (Probably will not need to use M)

Let's call this region the "cavity". Choose a spherical Gaussian Surface (S1) centered on O, with r < a.



On this surface, $\vec{E}[\vec{r}]$ = Ecavity[r] $\hat{r}$ and the differential area element is $\overrightarrow{dS1}$ = dS1 $\hat{r}$ (Carefully keep track of what is a vector and what is a scalar in all of the expressions.)

(We use the dummy variable rprime for the radius if it looks like we will be integrating over rprime.)

The charge density inside S1 is ρ[rprime] = 0; the differential volume element for this particular spherical symmetry (where ρ does not depend on θ or ϕ) is dτ1 = 4 π rprime^2 drprime

Gauss's Law for this surface is: $\int_{S1} \vec{E}[\vec{r}] \cdot d\overrightarrow{S1} = \frac{1}{\epsilon_0} \int_{\tau 1} \rho[\text{rprime}]\, d\tau 1$

OR: $\int_{S1} \text{Ecavity}[r]\ \hat{r}.dS1\, \hat{r} = \frac{1}{\epsilon_0} \int_{\tau 1} \rho[\text{rprime}]\, 4\pi\, \text{rprime}^2\, d\text{rprime}$

which can be rewritten: $\int_{S1} \text{Ecavity}[r] \left(\hat{r} \cdot \hat{r}\right) dS1 = \frac{1}{\epsilon_0} \int_{\tau 1} (0)\, 4\pi\, \text{rprime}^2\, d\text{rprime}$

First, $\hat{r} \cdot \hat{r}$ = 1. Second, on the spherical surface S1, r is a constant (equal to the radius of S1), so Ecavity[r] is constant and can come outside of the lhs integral. Third, the RHS = 0.

Therefore: Ecavity[r] * $\int_{S1} dS1$ = 0. $\int_{S1} dS1$ = the area of S1, a sphere of radius r, which is non-zero. (We ignore the possibility that this is not true at r = 0.) Therefore, for r < a, Ecavity[r] = 0. Since Ecavity[r] is the magnitude of the vector $\vec{E}[\vec{r}]$ inside the cavity, $\vec{E}[\vec{r}]$ = 0 for r < a.

As seen, M is of not much use for this part of the problem.

```
(* Input Cell -- you really don't need to use it *)
```

(e) Find $\vec{E}$inside_shell[$\vec{r}$], i.e., for a ≤ r ≤ b (Suggest you set up both sides of Gauss’s Law, then use M.)

```
this is a text cell – type in your analysis of the LHS and RHS of Gauss’s Law here:
```

We apply Gauss's Law to a spherical Gaussian surface S2 with radius r such that a ≤ r ≤ b (r inside the thick spherical shell)

Since the electric field everywhere in this geometry is constant on and perpendicular to any sphere centered on the origin, the LHS of Gauss's Law for S2, using the above symmetry arguments is simply: LHS = Eshell[r] * $\int_{S2} dS2$ = Eshell[r] * $4\pi r^2$.

(The integral $\int_{S2} dS2$ is simply the area of S2, a sphere of radius r, i.e., $4\pi r^2$.)

The RHS = $\frac{1}{\epsilon o}\int_{\tau 2}\rho[\text{rprime}]\, 4\pi\, \text{rprime}^2\, d\text{rprime}$, where τ2 is the volume consisting of the empty spherical cavity (rprime < a) plus the spherical shell contained between a ≤ rprime ≤ r.

As above, ρ[rprime] = 0 for rprime < a, so the empty cavity does not contribute to the RHS integral.

Thus, the remaining contribution is: RHS = $\frac{1}{\epsilon o}\int_a^r \rho_o\, \text{Exp}[-\alpha\, \text{rprime}]\, 4\pi\, \text{rprime}^2\, d\text{rprime}$, where again, $d\tau 2 = 4\pi\, \text{rprime}^2\, \text{drprime}$

We define both sides in M and then use Solve[ ] to find the function Eshell[r] which represents the magnitude of the non-zero component of E (the radial component). (Here, Solve[ ] is simply performing the trivial division by $4\pi r^2$.)

```
(* Input Cell *)
ClearAll["`*"]
ρ[rprime_] = ρo Exp[-α rprime];
RHS = 1/εo ∫_a^r ρ[rprime] 4 π rprime^2 ⅆrprime;
LHS = Eshell[r] * 4 π r^2 ;
sol = Solve[LHS == RHS, Eshell[r]];
Eshell[r_] = Eshell[r] /. sol[[1]] // Simplify;
Print["Eshell[r] = ", Eshell[r]]
```

$$\text{Eshell}[r] = \frac{1}{r^2\alpha^3\epsilon o} e^{-(a+r)\alpha}\left(e^{r\alpha}\left(2+2a\alpha+a^2\alpha^2\right)\rho o - e^{a\alpha}\left(2+2r\alpha+r^2\alpha^2\right)\rho o\right)$$

The vector field $\vec{E}$inside_shell[$\vec{r}$] is simply Eshell[r] $\hat{r}$ (the magnitude times the unit vector).

(f) Now find $\vec{E}$ outside the shell, $\vec{E}$outside_shell[$\vec{r}$], that is, $\vec{E}$ for r > b.

this is a text cell – type in your reasoning here

We apply Gauss's Law using a spherical Gaussian surface S3 with a radius r > b (r outside of all the charge in the thick spherical shell).

The LHS of Gauss's Law for S3, using the above symmetry arguments is : LHS = Eoutside_shell[r] * $\int_{S3} d S3$ = Eoutside_shell[r] $*$ 4 π r^2

(The integral $\int_{S3} d S3$ is simply the area of S3, which is a sphere of radius r, i.e., 4 π r^2.)

The RHS = $\frac{1}{\epsilon o} \int_{\tau 3} \rho[\text{rprime}]\, 4\, \pi\, \text{rprime}^2\, d\text{rprime}$, where $\tau 3$ is the volume consisting of the empty spherical cavity (rprime < a) plus the spherical shell contained between a ≤ rprime ≤ b plus the volume of the region for b ≤ rprime ≤ r

As above, ρ[rprime] = 0 for rprime < a, so the empty cavity does not contribute to the RHS integral. Likewise, ρ[rprime] = 0 for rprime > b, so the empty outer region does not contribute to the RHS integral.

Thus, the remaining contribution is: RHS = $\frac{1}{\epsilon o} \int_a^b \rho_o\, \text{Exp}[-\alpha\, \text{rprime}]\, 4\, \pi\, \text{rprime}^2\, d\text{rprime}$. Note the integral represents all the charge in the totality of the thick shell.

We again define both sides in M and then use Solve[ ] to find the function Eoutsideshell[r] which represents the magnitude of the non-zero component of E (the radial component). [Again, Solve[ ] is simply performing the trivial division by 4 π r^2]

```
(* Input Cell *)
RHS = 1/εo ∫_a^b ρ[rprime] 4 π rprime^2 ⅆrprime
LHS = Eoutsideshell[r] * 4 π r^2 ;
sol = Solve[LHS == RHS, Eoutsideshell[r]];
Eoutsideshell[r_] = Eoutsideshell[r] /. sol[[1]];
Print["Eoutsideshell[r] = ", Eoutsideshell[r]]
```

$$\frac{1}{\alpha^3\, \epsilon o} 4\, e^{-(a+b)\,\alpha}\, \pi \left(e^{b\,\alpha}\, (2 + a\,\alpha\, (2 + a\,\alpha)) - e^{a\,\alpha}\, (2 + b\,\alpha\, (2 + b\,\alpha))\right) \rho o$$

(g) Use If functions to define the magnitude of $\vec{E}$[r],(which I write as EE[r]), for *all* r (An alternative approach is to use PieceWise[ ].)

(ok, as a gift — this will work: `EE[r_] = If[r < a,0,If[a≤r≤b,Eshell[r],If[r > b,Eoutsideshell[r]]]]`)

Chose values for the constants and Plot EE[r] from r = 0 to some r > b.

(I used the following values: `a = 1; b = 2; α = 0.5; ρo = 1; ϵo = 1;`)

```
(* Input Cell *)
EE[r_] = If[r < a, 0, If[a ≤ r ≤ b, Eshell[r], If[r > b, Eoutsideshell[r]]]];
a = 1; b = 2; α = 0.5; ρo = 1; ϵo = 1;
plotEE = Plot[EE[r], {r, 0, 6}, AxesLabel → {"r", "Eradial[r]"}];
lines1 =
  Graphics[{Line[{{a, 0}, {a, 1}}], Line[{{b, 0}, {b, 1}}], Text["r = a", {a, 0.2},
     Background → LightRed], Text["r = b", {b, 0.2}, Background → LightGreen]}];
Show[plotEE, lines1]
```



Just for comparison, below is a plot of the Charge Enclosed (Qenclosed[r]) by a spherical Gaussian Surface of radius r for our particular ρ[r]

To beat a very dead horse, for a spherical symmetric ρ[r]: $\quad \text{EE[r]} *(4\,\pi\, r^2) = \frac{\text{Qenclosed}[r]}{\epsilon o}$

```
Clear[a , b , α , ρo , ϵo]
Qenclosed[r_] = If[r < a, 0, If[a ≤ r ≤ b,
    ∫_a^r ρ[rprime] 4 π rprime^2 ⅆrprime, If[ r > b, ∫_a^b ρ[rprime] 4 π rprime^2 ⅆrprime]]];
a = 1; b = 3; α = 0.5; ρo = 1; ϵo = 1;
plotQ = Plot[Qenclosed[r], {r, 0, 6}, AxesLabel → {"r", "Qenclosed[r]"}];
lines2 =
  Graphics[{Line[{{a, 0}, {a, 35}}], Line[{{b, 0}, {b, 35}}], Text["r = a", {a, 7},
     Background → LightRed], Text["r = b", {b, 7}, Background → LightGreen]}];
Show[plotQ, lines2]
```



The relation between this curve and the above plot of E[r] is: $E[r] = \dfrac{\text{Qenclosed}[r]}{4\,\pi\,\epsilon o\ r^2} = k\,\dfrac{\text{Qenclosed}[r]}{r^2}$

Note that the constant value of Qenclosed for r ≥ b equals the total charge contained in the whole shell which is:

$$\text{Qtotal} = \int_a^b \rho[\text{rprime}]\, 4\,\pi\,\text{rprime}^2\, d\,\text{rprime}$$

(h) Produce a VectorPlot of EE[r]·$\hat{r}$ in two dimensions (i.e., in a plane passing through the polar axis of the sphere). Remember, you will need to explicitly define 2D vector functions (for the three regions: r < a; a ≤ r ≤ b, and r > b) and convert them to Cartesian coordinates to use M’s VectorPlot function.

```
(* Input Cell *)
Clear[a , b , α , ρo , ϵo]
EcavityCartesian[x_, y_] = {0, 0}
EoutsideshellCartesian[x_, y_] =
 TransformedField["Polar" → "Cartesian", {Eoutsideshell[r], 0}, {r, θ} → {x, y}]
EshellCartesian[x_, y_] =
 TransformedField["Polar" → "Cartesian", {Eshell[r], 0}, {r, θ} → {x, y}]
EECartesian[x_, y_] = If[Abs[(x^2 + y^2)^(1/2)] < a, EcavityCartesian[x, y],
   If[a < Abs[(x^2 + y^2)^(1/2)] < b, EshellCartesian[x, y],
    If[Abs[(x^2 + y^2)^(1/2)] > b, EoutsideshellCartesian[x, y]]]];
a = 1; b = 3; α = 0.5; ρo = 1; ϵo = 1;
vecplotEE = VectorPlot[EECartesian[x, y], {x, -6, 6}, {y, -6, 6}];
spheres =
  Graphics[{Green, Circle[{0, 0}, a], Red, Circle[{0, 0}, b], Text["r = a", {a, -0.3},
     Background → LightRed], Text["r = b", {b, 0.3}, Background → LightBlue]}];
vecplotEEfinal = Show[vecplotEE, spheres]
```

{0, 0}

$$\left\{-\left(\left(e^{-(a+b)\alpha}\, x\,\left(2 e^{a\alpha} - 2 e^{b\alpha} + 2 b e^{a\alpha}\alpha - 2 a e^{b\alpha}\alpha + b^2 e^{a\alpha}\alpha^2 - a^2 e^{b\alpha}\alpha^2\right)\rho o\right) \Big/ \left(\left(x^2+y^2\right)^{3/2}\alpha^3\,\epsilon o\right)\right),\ -\left(\left(e^{-(a+b)\alpha}\, y\,\left(2 e^{a\alpha} - 2 e^{b\alpha} + 2 b e^{a\alpha}\alpha - 2 a e^{b\alpha}\alpha + b^2 e^{a\alpha}\alpha^2 - a^2 e^{b\alpha}\alpha^2\right)\rho o\right) \Big/ \left(\left(x^2+y^2\right)^{3/2}\alpha^3\,\epsilon o\right)\right)\right\}$$

$$\left\{\left(e^{-\left(a+\sqrt{x^2+y^2}\right)\alpha}\, x\left(e^{\sqrt{x^2+y^2}\,\alpha}\left(2+2 a\alpha + a^2\alpha^2\right)\rho o - e^{a\alpha}\left(2 + 2\sqrt{x^2+y^2}\,\alpha + \left(x^2+y^2\right)\alpha^2\right)\rho o\right)\right) \Big/ \left(\left(x^2+y^2\right)^{3/2}\alpha^3\,\epsilon o\right),\ \left(e^{-\left(a+\sqrt{x^2+y^2}\right)\alpha}\, y\left(e^{\sqrt{x^2+y^2}\,\alpha}\left(2+2 a\alpha + a^2\alpha^2\right)\rho o - e^{a\alpha}\left(2 + 2\sqrt{x^2+y^2}\,\alpha + \left(x^2+y^2\right)\alpha^2\right)\rho o\right)\right) \Big/ \left(\left(x^2+y^2\right)^{3/2}\alpha^3\,\epsilon o\right)\right\}$$



(i) Interpret the resulting 2D Vector Plot for the given spherically symmetric charge distribution:

$$\rho[r] = \begin{cases} 0 & \text{for } r < a \\ \rho o\ \text{Exp}[-\alpha\, r] & \text{for } a \le r \le b \\ 0 & \text{for } r > b \end{cases}$$

(this is a text cell - type in your interpretation)

From our symmetry arguments and Gauss's Law used to derive the E[r] functions for these three regions, we know that:

(1) In all regions, $\vec{E}$[r] = Er[r] $\hat{r}$. Consistent with this, the vector plot shows all E vectors are radial; at any given r, the LENGTH of $\vec{E}$[r] is the same.

(2) No charge is contained within spherical Gaussian surfaces with r < a, so all E vectors inside this cavity = $\vec{0}$. No vectors seen in the VectorPlot for this region because they are all of zero length.

(3) Inside the shell (a ≤ r ≤ b), the charge contained within spherical Gaussian surfaces starts at 0 from r = a and reaches a maximum at r = b. Consequently the magnitude of Er[r] increases with r from 0 at r = a to a maximum at r = b. The VectorPlot shows this increase.

(4) For r > b, the charge contained within spherical Gaussian surfaces of radius r remains fixed at

$$\text{Qtotal} = \int_a^b (\rho o\ \text{Exp}[-\alpha\ \text{rprime}])\left(4\,\pi\ \text{rprime}^2\right) d\,\text{rprime} =$$

$$\frac{1}{\alpha^3} 4\pi \left(e^{-a\alpha}(2 + a\alpha(2 + a\alpha)) + e^{-b\alpha}(-2 - b\alpha(2 + b\alpha))\right)\rho o$$

(5) From past work with E-Fields due to spherically symmetric charge distributions, we know that the field outside the charge distribution will be that of a point charge = Qtotal located at the origin. It should be a radial field whose magnitude falls as $\frac{1}{r^2}$. The VectorPlot indicates such a field.

(j) Now derive the electric potential V[r] for all regions. It is very easy to show that if $\vec{E}$[r] = Er[r] $\hat{r}$, V will be a function of r only. We will go with that.

`(this is a text cell - type in your interpretation)`

First, we copy and paste clean spherical coordinate versions of Er[r] (the magnitude of $\vec{E}$[r] = Er[r] $\hat{r}$) for the three regions:

	Eoutsideshell[r] =	$\frac{\left(e^{-a\alpha}(2+a\alpha(2+a\alpha))+e^{-b\alpha}(-2-b\alpha(2+b\alpha))\right)\rho o}{r^2\alpha^3\epsilon o}$	**for r > b**
Er[r] =	**Eshell[r] =**	$\frac{\left(e^{-a\alpha}(2+a\alpha(2+a\alpha))+e^{-r\alpha}(-2-r\alpha(2+r\alpha))\right)\rho o}{r^2\alpha^3\epsilon o}$	**for a ≤ r ≤ b**
	Ecavity[r] =	**0**	**for r < a**

We generally take the zero of potential at r = ∞, so that:

V[r] - V[∞] = – $\int_\infty^r$E[rprime] $d$rprime, where rprime, again, is a dummy variable. (This is a LINE INTEGRAL).

This expression gives the work per unit charge to move a test charge from infinity to the position r.

Just a reminder—this all holds for radial dependent functions. CAREFULLY NOTE: the – sign. If troubled by this, go here:

http://www.phys.ufl.edu/~acosta/phy2061/lectures/ElectricPotential.pdf

For $\vec{E}$[r] = Er[r] $\hat{r}$, we have proven that the only component of $\vec{E}$[r] that contributes to the line integral (and the potential) is the radial component; any motion at constant r does not require work.

FIRST r > b:

TO FIND Voutsideshell[r] (corresponding to r > b) we form the following integral:

Voutsideshell[r] = – $\int_\infty^r$Eoutside[rprime] $d$rprime,

where Eoutside[rprime] is the magnitude of the radial vector field.

```
(*  Input cell *)
Clear[a, α, b, ϵo, ρo]
```

$$\text{Eoutsideshell[r\_]} = \frac{1}{r^2\,\alpha^3\,\epsilon o}\left(e^{-a\,\alpha}\,(2 + a\,\alpha\,(2 + a\,\alpha)) + e^{-b\,\alpha}\,(-2 - b\,\alpha\,(2 + b\,\alpha))\right)\rho o;$$

```
(* a reminder *)
$Assumptions = (b ≤  rprime ≤ ∞) && rprime ∈ Reals && r ∈ Reals && r ≥ b;
Voutsideshell[r_] =
```

$$-\int_{\infty}^{r}\text{Eoutsideshell[rprime]}\,d\text{rprime}\quad (*\ \int_{\infty}^{r}\text{Eoutsideshell[r] is defined above } *)$$

$$\frac{1}{r\,\alpha^3\,\epsilon o}\left(e^{-a\,\alpha}\,(2 + a\,\alpha\,(2 + a\,\alpha)) + e^{-b\,\alpha}\,(-2 - b\,\alpha\,(2 + b\,\alpha))\right)\rho o$$

SECOND, inside the shell (a ≤ r ≤ b):

To find Vshell[r], we integrate Eoutside[r] from ∞ to b (which equals Voutsideshell[b]) and then integrate Eshell[r] from b to r (to get what I call Vinsideonly[r]). (Of course we insert the – sign.)

Vshell[r] is the *sum* of the two contributions. The limits on these integrals are consistent with moving the test charge from ∞ to an r inside the shell, a ≤ r ≤ b. Note: I have changed the assumptions to aid the Vinsideonly[r] integration. I recommend using Simplify near or at end.

Third, inside the empty cavity (r < a): Since there is no charge enclosed, the electric field in the cavity is zero and the extra work to move the test charge from r = a to anywhere in the empty cavity will also be zero. Thus, the potential inside the cavity Vcavity[r] = Vshell[a] (a constant!).

Here are the calculations to obtain Vshell[r]

```
(*  Input cell  *)
```

$$\text{Eshell[r\_]} = \frac{1}{r^2\,\alpha^3\,\epsilon o}\left(e^{-a\,\alpha}\,(2 + a\,\alpha\,(2 + a\,\alpha)) + e^{-r\,\alpha}\,(-2 - r\,\alpha\,(2 + r\,\alpha))\right)\rho o;$$

```
$Assumptions = ( a ≤  rprime ≤ b) && rprime ∈ Reals &&
   ( a ≤  r ≤ b) && r ∈ Reals && r > 0 && α ∈ Reals && α > 0;
```

$$\text{Vinsideshellonly[r\_]} = -\int_{b}^{r}\text{Eshell[rprime]}\,d\text{rprime}$$

```
Vshell[r_] = Voutsideshell[b] + Vinsideshellonly[r];
Vshell[r_] = % // Simplify
```

$$-\frac{1}{b\,r\,\alpha^3\,\epsilon o}e^{-(a+b+r)\,\alpha}\left(2\,b\,e^{(a+b)\,\alpha} - 2\,e^{(a+r)\,\alpha}\,r + b\,e^{(a+b)\,\alpha}\,r\,\alpha - b\,e^{(a+r)\,\alpha}\,r\,\alpha + e^{(b+r)\,\alpha}\,(-b + r)\,(2 + a\,\alpha\,(2 + a\,\alpha))\right)\rho o$$

$$-\frac{1}{r\,\alpha^3\,\epsilon o}e^{-(a+b+r)\,\alpha}\left(e^{(a+r)\,\alpha}\,r\,\alpha\,(1 + b\,\alpha) + e^{(a+b)\,\alpha}\,(2 + r\,\alpha) - e^{(b+r)\,\alpha}\left(2 + 2\,a\,\alpha + a^2\,\alpha^2\right)\right)\rho o$$

THIRD, inside the empty cavity (r < a): Since there is no charge enclosed, the electric field in the cavity is zero and the extra work to move the test charge from r = a to anywhere in the empty cavity will also be zero. Thus, the potential inside the cavity Vcavity[r] = Vshell[a] (a constant!).

```
(*  Input cell  *)
Vcavity[r_] = Vshell[a]
```

$$-\frac{1}{a\,\alpha^3\,\epsilon o}e^{-(2a+b)\alpha}\left(e^{(a+b)\alpha}(2+a\alpha)+a\,e^{2a\alpha}\alpha\,(1+b\alpha)-e^{(a+b)\alpha}\left(2+2a\alpha+a^2\alpha^2\right)\right)\rho o$$

(k) Now plot V[r] for all three regions combined.

```
(*  Input cell *)
V[r_] =
  If[r < a, Vcavity[r], If [a ≤ r ≤ b, Vshell[r], If[r > b, Voutsideshell[r]]]];
a = 1; b = 3; α = 0.5; ρo = 1; ϵo = 1;
lines3 = Graphics[{Line[{{a, 0}, {a, 1.4}}],
    Line[{{b, 0}, {b, 1.4}}], Text["r = a", {a, 0.7}, Background → LightRed],
    Text["r = b", {b, 0.7}, Background → LightGreen]}];
plotv = Plot[ V[r], {r, 0, 10}, PlotRange → {0, 1.5}, AxesLabel → {"r", "V[r]"}];
Show[plotv, lines3]
```



(l) Plot the magnitude of $\vec{E}$[r], V[r], the charge density ρ[r], and the Qenclosed[r] together in a single graph. You may want to adjust some of the amplitudes so they stay usefully scaled.

Plotting the magnitude of E[r], V[r], the charge density ρ[r], and the Qenclosed[r] over all the regions to compare their behavior (amplitudes have been adjusted to simplify comparison of curve behaviors):

```
(*  Input Cell *)
ρ[r_] = If[r < a, 0, If [a ≤  r ≤ b, ρo Exp[-α r], If[r > b, 0]]];
(* just some help *)
pp = Plot[{ EE[r], 0.2 V[r], 0.5 ρ[r], 0.01 Qenclosed[r]}, {r, 0, 10}, PlotRange → All,
   PlotLegends → {"Eradial[r]", "V[r]", "ρ[r]", "Qenclosed[r]"}];
lines4 = Graphics[{Line[{{a, 0}, {a, 0.4}}], Line[{{b, 0}, {b, 0.4}}],
    Text["r = a", {a, 0.1}, Background → LightRed],
    Text["r = b", {b, 0.1}, Background → LightGreen]}];
Show[pp, lines4]
```



(m) Make some interpretive remarks about these plots (worry not about repeating comments from above).

(this is a text cell - type in your interpretation)

Although in some cases repetitive, here are a few more interpretive comments:

Outside the shell (r > b), E[r] falls as $\frac{1}{r^2}$ and V falls as $\frac{1}{r}$ corresponding to the $\vec{E}$ and V of a point charge Qtotal located at the origin,

where Qtotal = $\int_a^b \rho[\text{rprime}]\, 4\,\pi\, \text{rprime}^2\, d\,\text{rprime}$

Qtotal is the constant value of Qenclosed for r > b.

Inside the shell (a ≤ r ≤ b), Qenclosed[r] increases from zero (at r = a) to Qtotal (at r = b), due to the integration of the Exp[-α r] factor in ρ[r]; the E field increases in magnitude accordingly. Because there is no charge contained in the empty cavity, Ecavity = 0.

As r decreases from r = b towards r = a, one must do work to move a test charge because there continues to be charge inside the corresponding Gaussian surface (and thus a non-zero radial electric field). Thus, V[r] continues to increase with decreasing r. Once r ≤ a, no more work is needed to move the test charge.

Ecavity = 0; Vcavity is a constant.

Both E and V are continuous at r = a and r = b. Such continuity is consistent with the Boundary Conditions of E and V at an uncharged interface. (The spherical shell with nonzero ρ has no net free charge attached to it's inner and outer surfaces.)

The BCs relevant here are:

- $E^{\perp}_{r<a} - E^{\perp}_{r>a} = \frac{\sigma[a]}{\epsilon 0}$ (where $\sigma[a] = 0$)
- $E^{\perp}_{r<b} - E^{\perp}_{r>b} = \frac{\sigma[b]}{\epsilon 0}$ (where $\sigma[b] = 0$)
- $V_{r<a} = V_{r>a}$, and
- $V_{r<b} = V_{r>b}$

(n) Generate a 2D Contour Map of the potential over all three regions. {Hint: ContourPlot uses Cartesian Coordinates—you can use your general V[r] above and simply replace it with V[r] /. r → $(x^2 + y^2)^{\frac{1}{2}}$}

Using Show, superimpose this ContourPlot with the VectorPlot of EE[r]·$\hat{r}$ above.

I am rewriting and re-executing vecplotEE below to spiff up the final plot.

```
(*  Input Cell *)
vecplotEE =
  VectorPlot[EECartesian[x, y], {x, -6, 6}, {y, -6, 6}, VectorColorFunction → Hue];
spheres = Graphics[{Green, Circle[{0, 0}, a], Red, Circle[{0, 0}, b],
    Text["r = a", {a, -0.3}, Background → LightRed],
    Text["r = b", {b, 0.3}, Background → LightBlue]}];
vecplotEEfinal = Show[vecplotEE, spheres];
conplotV =
  ContourPlot[V[r] /. r → (x^2 + y^2)^(1/2), {x, -6, 6}, {y, -6, 6}, Contours → 30];
Show[conplotV, vecplotEEfinal]
```



(o) Interpret your combined plot.

(this is a text cell - type in your interpretation)

Reminder, the 2D plot contour is on a plane that cuts through the z axis of the original problem.

The charge density (ρ[r] = ρo Exp[-α r]) is non-zero only between r = a and r = b.

- **Due to the symmetry of ρ[r], $\vec{E}$ is radial: $\vec{E}$[$\vec{r}$] = E[r] $\hat{r}$ for all r. Since no work is done moving along a surface of constant r (E remains constant), the surfaces of constant V are spheres (here, circles).**

Note : IF you point your mouse at any contour in the plot, the value of V along that contour will pop up.

• Brain-check: V is constant (and non-zero) in all of the region where r < a. Because the increment in V between contours is NOT infinitesimal, you cannot see any contours between r = a and the first plotted contour, which appears at larger r. BUT V is changing (here decreasing) between r = a and the first plotted contour.

• Since $\vec{E}[\ \vec{r}\ ] = -$ the gradient of V[r], which is in the direction of steepest descent in the potential, we see that all of the E vectors are perpendicular to the contours.